

Project 1 - Mark Making

Due Next Week

Use the Processing Development Environment to create an interactive mark making application (drawing program). You may choose to emulate traditional mark making tools (pen, brush, stamp, marker, etch-a-sketch) or you can stretch the definition of mark making.

The primary requirement is that the user is able to discern an immediate correlation between their input and the program's reaction.

Remember that Processing will leave previously drawn elements on screen unless the `background()` function is called or if a new element is drawn over the existing images. This can be a simple method to visually record a history of the users interaction, however it is quite limited in that Processing retains no knowledge of this history. You can record the users input in variables, most likely using an array. Once the history is stored in an array, it can be retrieved as necessary and displayed on screen appropriately.

Here's an example sketch (I'd hope that you get more creative):

```
int num = 50;
int x[] = new int[num];
int y[] = new int[num];

void setup() {
  size(300,300);
  noStroke();
  smooth();
  frameRate(30);
  fill(255,102);
}

void draw() {
  background(51);

  //shift the values to the right
  //removing the oldest value
  for(int i = num-1; i>0;i--){
    x[i] = x[i-1];
    y[i] = y[i-1];
  }

  //add new values to the beginning of the array
  x[0] = mouseX;
  y[0] = mouseY;

  //draw ellipses at each position in the array
  for(int i=0;i<num;i++){
    ellipse(x[i],y[i],i/2,i/2);
  }
}
```

